

Slides for Chapter 10: Toward Applied General-Equilibrium (aka CGE) Modeling

Copyright: James R. Markusen University of Colorado, Boulder

10.1 CES functions and the calibrated share form: List of symbols

α : Distribution parameter

θ : Cost share

ρ : CES elasticity parameter

σ : Elasticity of substitution

A : Scaling factor

C : Cost function

K : Input of capital

L : Input of labor

P : Price

X : Output

$L^0, K^0, X^0, P_L^0, P_K^0, P_X^0$: Initial observed values of variables

Textbook form of production function:

$$X = A(\alpha K^\rho + (1 - \alpha)L^\rho)^\rho \quad -\infty \leq \rho \leq 1$$

$$X = A \left(\alpha K^{\frac{\sigma-1}{\sigma}} + (1 - \alpha)L^{\frac{\sigma-1}{\sigma}} \right)^{\frac{\sigma}{\sigma-1}} \quad \sigma = \frac{1}{1 - \rho}$$

Calibrated share form of production function

$$X = X^0 \left(\theta_K \left(\frac{K}{K^0} \right)^{\frac{\sigma-1}{\sigma}} + \theta_L \left(\frac{L}{L^0} \right)^{\frac{\sigma-1}{\sigma}} \right)^{\frac{\sigma}{\sigma-1}}$$

where $\theta_K = \frac{P_K^0 K^0}{P_X^0 X^0}$ $\theta_L = \frac{P_L^0 L^0}{P_X^0 X^0}$ remain fixed after calibration

Textbook form of cost function:

$$C(P, X) = \frac{X}{A} \left(\alpha^\sigma P_K^{1-\sigma} + (1-\alpha)^\sigma P_L^{1-\sigma} \right)^{\frac{1}{1-\sigma}}$$

Calibrated share form:

$$C(P, X) = C(P^0, X^0) \frac{X}{X^0} \left(\theta_K \left(\frac{P_K}{P_K^0} \right)^{1-\sigma} + \theta_L \left(\frac{P_L}{P_L^0} \right)^{1-\sigma} \right)^{\frac{1}{1-\sigma}}$$

Textbook form of factor demand function:

$$L = \left(\frac{1 - \alpha}{P_L} \right)^\sigma C(P, X)$$

Calibrated share form:

$$L = L^0 \left(\frac{X}{X^0} \right) \left(\frac{P_X P_L^0}{P_L P_X^0} \right)^\sigma = \left(\frac{P_L^0 L^0}{P_X^0 X^0} \right) \left(\frac{P_L^0}{P_X^0} \right)^{\sigma - 1} \left(\frac{P_X}{P_L} \right)^\sigma X$$

$$L = \theta_L \left(\frac{P_L^0}{P_X^0} \right)^{\sigma - 1} \left(\frac{P_X}{P_L} \right)^\sigma X$$

As prices change, the shares of L and K in output change

$$\theta_L^* = \frac{P_L L}{P_X X} = \left(\frac{L^0}{X^0} \right) \left(\frac{P_L^0}{P_X^0} \right)^\sigma \left(\frac{P_X}{P_L} \right)^{\sigma-1}$$

$$\theta_L^* = \left(\frac{P_L^0 L^0}{P_X^0 X^0} \right) \left(\frac{P_L^0}{P_X^0} \right)^{\sigma-1} \left(\frac{P_X}{P_L} \right)^{\sigma-1}$$

$$\theta_L^* = \theta_L \left(\frac{P_X/P_X^0}{P_L/P_L^0} \right)^{\sigma-1}$$

where θ_L is the initial (fixed) calibrated share

10.2 The MPS/GE subsystem of GAMS

GAMS includes a higher-level language, written by Rutherford, MPS/GE: mathematical programming system for general equilibrium. MPS/GE uses the MCP solver in GAMS.

First, the program has routines for calibrating and writing all constant-returns CES and CET functions, up to three levels of nesting.

Second, and closely related, the form of the data required to specify a CES/CET function is exactly the data modelers have, so there is a swift and easy move from an accounting matrix to the calibration of the model.

Third, a lot of market-clearing and income-balance equations are written automatically by MPS/GE so the modeler doesn't have to worry about doing so.

Fourth, and closely related, a lot of errors that can occur when a modeler writes out his or her equations cannot occur in MPS/GE.

If there is a tax or markup, for example, the revenues must be assigned to some agent and will be allocated automatically to that agent by the income-balance properties of the coding.

A few key words.

SECTOR (ACTIVITY)

Production activities that convert commodity inputs into commodity outputs. The variable associated with a sector is the activity (output) level.

For every sector/activity, mps/ge constructs a cost function and zero-profit inequality.

COMMODITY (MARKETS)

A good or factor. The variable associated with a commodity is its price, not its quantity.

For every commodity/market, mps/ge constructs a market-clearing inequality, using Shepard's lemma for good / factor demands

CONSUMERS

Individuals who supply factors and receive tax revenues, markups, and pay subsidies. In imperfectly competitive models, firm owners can be designated as consumers.

A government that receives tax revenue and buys public goods is also designated as a consumer. The variable associated with a consumer is income from all sources.

AUXILIARY

Additional variables, such as markup formulae or taxes with endogenous values which are functions of other variables such as prices and quantities. Please note the spelling of auxiliary: mistakes cause MPS/GE to crash, and you won't know why.

CONSTRAINT

An equation that is typically used to set the value of an auxiliary variable. In these appendix programs, constraint equations will be used to set the values of markups, which are auxiliary variables.

Here is what an MPS/GE program, embedded in a GAMS file, looks like, where the model name is M10-1.

GAMS statements such as declaring sets, parameters, parameter values, etc.

**** now control is passed to the MPS/GE subsystem ****

\$ONTEXT [this tells the GAMS compiler to ignore what follows, but the MPS/GE compiler will recognize the model statement that follows and will begin to pay attention]

\$MODEL: M10_1

Declare sectors, commodities, consumers, auxiliary variables

Production Blocks

Demand Blocks

Constraint equations

```
$OFFTEXT [control is passed back to GAMS]
```

```
**** now we are back in GAMS ****
```

```
$SYSINCLUDE MPSGESET M10_1
```

GAMS statements such as setting starting values of variables,
other parameter values, etc.

```
$INCLUDE M10_1.GEN  
SOLVE M0_1 USING MCP;
```

GAMS statements processing output

Below, we formulate problem introduced above as M3-4b using
MPS/GE

\$TITLE Model M10-2: TWOxTWOxONE economy -- MPS/GE version of model M3-4b

\$ONTEXT

This is the exact same model as M3-4b.GMS but uses the MPS/GE format.

<i>Markets</i>	<i>Production Sectors</i>			<i>Consumers</i>
	<i>X</i>	<i>Y</i>	<i>W</i>	<i>CONS</i>
<i>PX</i>	100		-100	
<i>PY</i>		100	-100	
<i>PW</i>			200	-200
<i>PL</i>	-25	-75		100
<i>PK</i>	-75	-25		100

\$OFFTEXT

PARAMETERS

TX ad-valorem tax rate for X sector inputs
 TY ad-valorem tax rate for Y sector inputs
 LENDOW labor endowment multiplier
 KENDOW capital endowment multiplier;

```
TX = 0; TY = 0;
LENDOW = 1;
KENDOW = 1;
```

```
$ONTEXT
```

```
$MODEL:M10_2
```

```
$SECTORS:
```

```
    X      ! Activity level for sector X
    Y      ! Activity level for sector Y
    W      ! Activity level for sector W (Hicksian welfare index)
```

```
$COMMODITIES:
```

```
    PX      ! Price index for commodity X
    PY      ! Price index for commodity Y
    PL      ! Price index for primary factor L
    PK      ! Price index for primary factor K
    PW      ! Price index for welfare (expenditure function)
```

```
$CONSUMERS:
```

```
    CONS    ! Income level for consumer CONS
```

\$PROD:X s:1

O:PX Q:100

I:PL Q:25 A:CONS T:TX

I:PK Q:75 A:CONS T:TX

\$PROD:Y s:1

O:PY Q:100

I:PL Q:75 A:CONS T:TY

I:PK Q:25 A:CONS T:TY

\$PROD:W s:1

O:PW Q:200

I:PX Q:100

I:PY Q:100

\$DEMAND:CONS

D:PW Q:200

E:PL Q:(100*LENDOW)

E:PK Q:(100*KENDOW)

\$OFFTEXT

\$SYSINCLUDE mpsgeset M10_2

PW.FX = 1;

```
$INCLUDE M10_2.GEN
```

```
SOLVE M10_2 USING MCP;
```

```
*           Solve the counterfactuals
```

```
TX = 0.5;
```

```
$INCLUDE M10_2.GEN
```

```
SOLVE M10_2 USING MCP;
```

```
TX = 0.5;
```

```
TY = 0.5;
```

```
$INCLUDE M10_2.GEN
```

```
SOLVE M10_2 USING MCP;
```

```
TX = 0;
```

```
TY = 0;
```

```
LENDOW = 2;
```

```
$INCLUDE M10_2.GEN
```

```
SOLVE M10_2 USING MCP;
```

```
LENDOW = 2;
```

```
KENDOW = 2;
```

```
$INCLUDE M10_2.GEN
```

```
SOLVE M10_2 USING MCP;
```


Now some more details.

(1) Production blocks

The terminology here is a bit confusing, since MPS/GE takes the information in a production block and generates a cost function, not a production function.

But the variable complementary with a production block (cost function) is an activity level. Let's take an example from the above program, adding the price field.

```
$PROD:Y   S:1
          O:PY       Q:100     P:1
          I:PL       Q: 75     P:1
          I:PK       Q: 25     P:1
```

First line

Name of activity (Y), value of substitution (here s:1) and transformation elasticities if there are several outputs.

Default elasticity of substitution is 0 (not 1!).

First column

Names of commodity outputs (O:) and inputs (I:).

Second column

Reference commodity quantities (Q:) – used for calibration.
Default = 1 if none specified.

Third column


Reference commodity prices (P) – used for calibration. Default = 1 if none specified, which is why they are omitted in the program above.

MPS/GE then takes this information to construct a cost function and, as a feature of CES functions, it is globally defined by a single reference point.

Think of putting an isoquant labeled 100 units of output, with elasticity of substitution 1, through input points $L = 75$, $K = 25$, with slope $PL/PK = 1$. That is what MPS/GE does for you.

In this simple case, it constructs the cost function:

$$100*(PL^{**.75})*(PK^{**.25}) = G = 100*PY;$$



Infor given in \$PROD block

value and quantity of output $X = 100$
 value and quantity of L = 75
 value and quantity of K = 25
 slope of isoquant at reference point = 1
 curvature of CES, $s = 1$



this info used by MPS/GE to construct cost function and pricing

The saving from using MPS/GE might not seem like a big deal, but believe me with many inputs, different prices for all inputs, and an elasticity of substitution of 3.5, it is a huge saving indeed.

One example of the treatment of taxes (others will follow later, including those with endogenous rates) is in the production block for X.

```
$PROD:X  s:1
          O:PX  Q:100
          I:PL  Q:25   A:CONS  T:TX
          I:PK  Q:75   A:CONS  T:TX
```

The “A” field means “assign” the revenue from tax TX to the agent CONS. Read it as the statement “assign to agent CONS the revenue from tax rate TX on inputs L and K”.

A utility function is also represented by a production block; that is, utility is a good which is produced from commodity inputs (including possibly factor inputs such as leisure).

Here is the utility function (W for welfare), in which utility (good PW) is produced from inputs of X and Y . MPS/GE constructs the underlying expenditure (cost) function.

```
$PROD:W  s:1
          O:PW    Q:200
          I:PX    Q:100
          I:PY    Q:100
```

A consumer's income constraint is also represented by a "block" in this case called a demand block.

In what follows, the consumer demands the utility good PW (the “D” field), and receives income from endowments (the “E” fields) of labor and capital.

MPS/GE automatically handles tax revenue or subsidy payments in the background, adding or subtracting them to the consumer's endowment income.

\$DEMAND:CONS

D:PW Q:200

E:PL Q:(100*LENDOW)

E:PK Q:100

MPS/GE also automatically looks after the market clearing conditions.

When the MPS/GE compiler sees “PL” listed under “commodities” it will construct a supply \geq demand inequality.

(1) find all sources of PL supply (outputs or endowments).
In our case, there is only one, in \$DEMAND: CONS

(2) find all sources of PL demand (inputs). In our case, there are in two, \$PROD: X and \$PROD: Y

(3) MPS/GE applies Shepard’s lemma to all cost functions to get the demands for PL in the PROD blocks.

(4) constructs the supply/demand inequality.

4. Example of how the Q and P fields are used to construct the underlying cost function.

All constant returns CES and CET functions can be completely characterized by a single point consisting of (1) input quantities (2) output quantities (3) input and output prices, and (4) the elasticity of substitution or transformation (there may be several levels of substitution elasticities).

MPS/GE constructs the underlying cost function from such a single observation. It is particularly important to specify the reference prices correctly. Consider the two production blocks:

```
$PROD: X      s:1
O:PX  Q:100   P:1
I:PL  Q: 25   P:1
I:PK  Q: 75   P:1
```

```

$PROD: X      s:1
O:PX  Q:100   P:1
I:PL  Q: 25   P:2
I:PK  Q: 75   P:0.667

```

These are both Cobb-Douglas production functions that can produce 100 X from inputs of 25 labor and 75 capital, and in each case the value of the inputs equals the value of the output.

But the isoquant has a different slope through that input combination in the two cases: the marginal rate of substitution in the first case is one, but 3 in the second case. Thus these are not the same technologies.

Later, with various taxes, it is important to divide values into price and quantity components when all prices cannot be normalized to one. Consider the following two production blocks.

```
$PROD: X      s:1  
O:PX Q:100    P:1  
I:PL Q: 25    P:1  
I:PK Q: 75    P:1
```

```
$PROD: X      s:1  
O:PX Q:100    P:1  
I:PL Q: 50    P:0.5  
I:PK Q: 75    P:1
```

In both cases, the values of inputs and output are the same, and would appear the same in the data matrix. But these are not the same functions, the first being a clearly more efficient technology than the second.

While both will have the same share parameters on L and K (0.25 and 0.75 respectively), the first technology will have a higher multiplicative “efficiency” parameter scaling up the output from given inputs (or scaling down the cost of output at given factor prices).

MPS/GE will automatically calculate this scaling parameter.

10.3 Armington formulation

Another feature of real data that confronts modelers is that there is generally two-way trade in any “good” in the data; that is, each good is both imported and exported.

This is often attributed to the fact that data is actually classified by industries and any industry is actually an aggregation of many goods.

Yet two-way trade or “cross hauling” will persist in the data even at an extremely fine level of disaggregation.

No research will ever get data that is free of cross hauling so the question is what to do about it.

	Production Sectors				Consumer	
Markets	X1	X2	E	M	W	CONS
P1	150		-100	50	-100	
P2		50	-25	75	-100	
PL	-100	-20				120
PK	-50	-30				80
PW					200	-200
PFX			125	-125		

Each “good” (industry) is both imported and exported. One way to handle this is to simply net out the two way trade from the gross flows and replace the data with net trade only.

If we do this in the present model, we arrive exactly a model essentially the same as M8-1. We can then proceed as before.

The other alternative in competitive models is to assume that the domestic and foreign goods are not really identical. This is the so-called Armington assumption: domestic and foreign goods in the same industrial classification are imperfect substitutes.

```

$PROD:W   s:1   G1:ESUB   G2:ESUB
          O:PW   Q:200
          I:P1   Q: 50   G1:
          I:PF1  Q: 50   G1:
          I:P2   Q: 25   G2:
          I:PF2  Q: 75   G2:

```

The domestic (P1) and foreign good (PF1) in industry 1 are in a nest as are the domestic and foreign goods in industry 2 and these nest have the same elasticity of substitution: a domestic and foreign good in the same industry have the same substitution elasticity regardless of industry.

The composite industry 1 good and the composite industry 2 good are then combined in an upper level nest.

This is generally assumed to have a lower elasticity of substitution: two industries are poor substitutes than a domestic and foreign good in the same industry.

Here we specify the upper nest with an elasticity of substitution of 1, while the domestic-foreign goods have an elasticity of substitution of $ESUB = 4$.

The rest of the program is rather familiar, except that the four trade activities are now defined for four distinct goods.

There is no need now to worry about a domestic good being both imported and exported. It cannot, by definition, be imported

\$TITLE Model M10-3: Small open economy model with an Armington
* *formulation.*

\$ONTEXT

	<i>Production Sectors</i>					<i>Consumer</i>	
<i>Markets</i>	<i>X1</i>	<i>X2</i>	<i>E</i>	<i>M</i>	<i>W</i>	<i>CONS</i>	
<i>P1</i>	150		-100	50	-100		
<i>P2</i>		50	-25	75	-100		
<i>PL</i>	-100	-20				120	
<i>PK</i>	-50	-30				80	
<i>PW</i>					200	-200	
<i>PFX</i>			125	-125			

\$OFFTEXT

PARAMETERS

PE2 Export price of good 2,
 PM1 Import price of good 1,
 PE1 Export price of good 1,
 PM2 Import price of good 2,
 TM2 Import tariff for good 2,
 ESUB Armington elasticity of substitution;

```
PE1 = 1;  
PM2 = 1;  
PE2 = 1;  
PM1 = 1;  
TM2 = 0;  
ESUB = 20;
```

```
$ONTEXT
```

```
$MODEL:M10_3
```

```
$SECTORS:
```

```
    X1      ! Production index for good 1  
    X2      ! Production index good 2  
    E1      ! Export index for good 1  
    E2      ! Export index for good 2  
    M1      ! Import index for good 1  
    M2      ! Import index for good 2  
    W       ! Welfare index
```

```
$COMMODITIES:
```

```
    P1      ! Price index for good 1  
    P2      ! Price index for good 1  
    PF1     ! Price index for imported good 1  
    PF2     ! Price index for imported good 2
```

PFX *! Read exchange rate index*
PW *! Welfare price index*
PL *! Wage index*
PK *! Capital rental index*

\$CONSUMERS:

CONS *! Income level for representative agent*

** Cobb-Douglas production in both sectors:*

\$PROD:X1 s:1

O:P1 *Q:150*
I:PL *Q:100*
I:PK *Q:50*

\$PROD:X2 s:1

O:P2 *Q:50*
I:PL *Q:20*
I:PK *Q:30*

** We scale the export price for good 1 and the import price*
** for good 2 to both be unity:*

\$PROD:E1

O:PFX *Q:(PE1*100)*
I:P1 *Q:100*

\$PROD:E2

O:PFX Q:(PE2*25)

I:P2 Q:25

\$PROD:M1

O:PF1 Q:50

I:PFX Q:(PM1*50)

\$PROD:M2

O:PF2 Q:75

I:PFX Q:(PM2*75) A:CONS T:TM2

\$PROD:W s:1

G1:ESUB

G2:ESUB

O:PW Q:200

I:P1 Q:50 G1:

I:PF1 Q:50 G1:

I:P2 Q:25 G2:

I:PF2 Q:75 G2:

\$DEMAND:CONS

D:PW Q:200

E:PL Q:120

E:PK Q:80

```
$OFFTEXT  
$SYSINCLUDE mpsgeset M10_3
```

```
PW.FX = 1;
```

```
M10_3.ITERLIM = 0;  
$INCLUDE M10_3.GEN  
SOLVE M10_3 USING MCP;  
M10_3.ITERLIM = 2000;
```

```
TM2 = 0.05;
```

```
$INCLUDE M10_3.GEN  
SOLVE M10_3 USING MCP;
```

```
TM2 = 0.10;
```

```
$INCLUDE M10_3.GEN  
SOLVE M10_3 USING MCP;
```

10.4 From an Input-Output Table into GAMS

In this chapter, we consider the problem of moving from real data to an MPS/GE computable model.

At the moment, we will consider data that is initially micro consistent, highly unlikely in real data (see M5-4).

Please note that the examples here are highly stylized, and you will likely need to work carefully with real data in order to understand precisely how it is organized.

One common form of data presentation is the IO (input-output) table. A typical format might be the following.

Typical structure of input-output data

	Agriculture fishery, energy	Mineral and metal products	Other industry	Business services	Gov services	House cons	Gov cons	Gross invest	Exports	Imports	Import tariffs
	1	2	3	4	5	6	7	8	9	10	11
Agriculture, fish, energy	Intermediate use (demand) matrix					Final demand			Net exports = (9) <i>minus</i> (10 + 11)		
Mineral and metal products											
Other industry											
Business services											
Government services											
Wage payments	Value added										
Gross oper surplus	Taxes and Subsidies										
Production taxes minus subs											

Column 1 sum
can be thought of as the total cost of production

(inputs into the column sector)

Row 1 sum 1-9 *minus* (10 + 11)
can be thought of as the total value of production

(destination of *outputs* from the row sector)

Final demand
+ net exports (X-M)
=
Value added + taxes

C + I + G + (X-M)
=
Wages + Rents + Tax + Tariffs

Probably not designed by a modeler: note problem of signs. Not simple to figure out what exactly micro-consistency means in this format.

In a more complete setting, there would be a second matrix breaking down import into what sectors they went into, intermediate and final demand.

We will just assume that they go into final demand.

Put some numbers into this. Again, we will assume micro-consistency, probably not something true in real data.

One real-world twist is thrown in: Trade does not balance, there is a trade surplus of 5. How to model this in GAMS and MPS/GE is discussed below.

Stylized IO table: IO-1

		Ag	Man	Ser	Cons	Exports	Imports	Row sums
		1	2	3	4	5	6	(1)-(5) minus (6)
Ag	1	5	6	9	30		5	45
Man	2	10	15	20	60	10		115
Ser	3	12	13	5	110			140
Wage payments	4	9	30	66				105
Gross oper surplus	5	9	51	40				100
Column sums		45	115	140	200	10	-5	

wages	105	
rents	100	
income	205	
=		
consump	200.0	
trade sur	5.0	
205.0		

An alternative way of representing the same economy is to use what is known as a SAM: social accounting matrix.

In this format, the economy is divided into “sectors” where we will see the term is used differently from the way we have used it.

Production, wages, rents, consumption, trade are all called “sectors”. All entries are positive values.

For a given sector, the column for that sector is the expenditures of the sector on other sectors.

For a given sector, the row is the receipts of that sector from other sectors.

By definition, the row sum for a given sector must equal the column sum.

SAM equivalent of IO table IO-1

Column: Expenditures on

Row: Receipt from		Ag	Man	Ser	Wages	Rents	Cons	Capital account	ROW	Row sums
		1	2	3	4	5	6	7	8	
Ag	1	5	6	9			30			50
Man	2	10	15	20			60		10	115
Ser	3	12	13	5			110			140
Wages	4	9	30	66						105
Rents	5	9	51	40						100
Cons	6				105	100				205
Capital account	7						5			5
ROW	8	5						5		10
		50	115	140	105	100	205	5	10	

Now let us see how we could translate the IO table or SAM into the way we have been organizing data for GAMS and MPS/GE.

We have to decide what is a sector and what is a market.

We have to decide which of the (positive) numbers of the IO table enter as a positive value in our GAMS version and which enter as negative.

One convention we adopt is to just net out, own-industry use of a good. So, for example, in the IO table, 50 units of Ag are produced, but 5 go into Ag.

So we will could just let final output be given by 45, the net output.

But in the MPS/GE file itself, we will allow substitution, so we will list the output as 50 with an own input of 5.

We file M10-4a.gms is below. It is designed to show clearly how the data in the IO table translate into the type of matrix that we have been using repeated for GAMS, now MPS/GE.

The one twist is the trade surplus. To maintain general-equilibrium consistency, assume that the representative consumer is endowed with -5 units of foreign exchange.

For example, the country has an outstanding debt of 5 units of FX which must be repaid. Income balance will occur with a trade surplus of 5, paying off the debt. Note the consumer's negative "endowment" of foreign exchange.

Counterfactuals in M10-4a are to double the world price of manufactures, then return that price to one and double the price of agriculture.

In both counterfactuals, the economy stops producing one good. This is due to the fact that there are more goods and factors and so there will be “flats” in the production frontier.

This is a well-known theory problem and will be explain some in lectures.

There are two common techniques to dampen the responsiveness of the economy to world price changes.

\$TITLE: M10-4a.gms

*CALIBRATES MODEL TO SHEET IO-1 IN M10-IOTABLE.XLS

*TRADE SURPLUS OF 5

*NETS OUT CROSS-HAULING: domestic and foreign goods perfect substitutes

*actual MAN exports 20, imports 10; AGR exports 5, imports 10

\$ONTEXT

		Ag	Man	Ser	Cons	Exports	Imports		
		1	2	3	4	5	6	UM(1:5) - (6)	
Ag	1	5	6	9	30		5	45	
Man	2	10	15	20	60	10		115	
Ser	3	12	13	5	110			140	
Wages	4	9	30	66				105	
Rents	5	9	51	40				100	
		<hr/>							
		45	115	140	200	10	-5		

Micro consistent: total value added (205 = 105+100) =
consumption + trade surplus (205 = 200 + 10 - 5)

	<i>Net Ag</i>	<i>Net Man</i>	<i>Net Ser</i>	<i>Export</i>	<i>Imports</i>	<i>Welf</i>	<i>Cons</i>	<i>Row</i>
<i>Ag</i>	40	-6	-9		5	-30		0
<i>Man</i>	-10	100	-20	-10		-60		0
<i>Ser</i>	-12	-13	135			-110		0
<i>Wages</i>	-9	-30	-66				105	0
<i>Rents</i>	-9	-51	-40				100	0
<i>For ex</i>				10	-5		-5	0
<i>Welfare</i>						200	-200	0
<i>Col sums</i>	0	0	0	0	0	0	0	

\$OFFTEXT

PARAMETERS

PWA world price of agriculture

PWM world price of manufactures;

PWA = 1;

PWM = 1;

\$ONTEXT

\$MODEL:M10_4

\$SECTORS:

AGR
MAN
SER
IMAGR EXAGR
EXMAN IMMAN
WELFARE

\$COMMODITIES:

PAGR
PMAN
PSER
PL
PK
PFX
PWEL

\$CONSUMERS:

CONS

\$PROD:AGR s:1

O:PAGR Q:45
I:PAGR Q:5
I:PMAN Q:10
I:PSER Q:12
I:PL Q:9
I:PK Q:9

\$PROD:MAN s:1

O:PMAN Q:115

I:PAGR Q:6

I:PMAN Q:15

I:PSER Q:13

I:PL Q:30

I:PK Q:51

\$PROD:SER s:1

O:PSER Q:140

I:PAGR Q:9

I:PMAN Q:20

I:PSER Q:5

I:PL Q:66

I:PK Q:40

\$PROD:IMAGR

O:PAGR Q:5

I:PFX Q:(5*PWA)

\$PROD:EXMAN

O:PFX Q:(10*PWM)

I:PMAN Q:10

\$PROD:EXAGR

O: PFX Q: (5*PWA)
I: PAGR Q: (5*1.001)

\$PROD: IMMAN

O: PMAN Q: (10*0.999)
I: PFX Q: (10*PWM)

\$PROD: WELFARE s: 1

O: PWEL Q: 200
I: PAGR Q: 30
I: PMAN Q: 60
I: PSER Q: 110

\$DEMAND: CONS

D: PWEL
E: PL Q: 105
E: PK Q: 100
E: PFX Q: (-5)

\$OFFTEXT

\$SYSINCLUDE MPSGESET M10_4

PWEL.FX = 1;
EXAGR.L = 0;
IMMAN.L = 0;

```
M10_4.ITERLIM = 0;  
$INCLUDE M10_4.GEN  
SOLVE M10_4 USING MCP;
```

```
M10_4.ITERLIM = 2000;  
$INCLUDE M10_4.GEN  
SOLVE M10_4 USING MCP;
```

** counterfactual: double the world price of the export good M*

```
PWM = 2;
```

```
$INCLUDE M10_4.GEN  
SOLVE M10_4 USING MCP;
```

** counterfactual: double the world price of the import good A*

```
PWM = 1;
```

```
PWA = 2;
```

```
$INCLUDE M10_4.GEN  
SOLVE M10_4 USING MCP;
```

The first, modeled in M10-4b, is to impose an Armington assumption and assume domestic and foreign goods are imperfect substitutes.

Suppose that the data shown earlier nets out two-way trade, and that the actual gross trade is

Ag:	export 5 units	import 10 units
Man:	export 20 units	import 10 units.

The elasticity of substitution between domestic and foreign goods in Ag and Man is set at 5.0.

Compare the results of the same two counter-factuals in M10-4b to those in M10-4a, and note that the production responses are much more dampened in M10-4b.

\$TITLE: M10-4b.gms

*CALIBRATES MODEL TO SHEET IO-1 IN M10-IOTABLE.XLS

*TRADE SURPLUS OF 5

*cross-hauling modeled with Armington assumption

\$ONTEXT

		Ag	Man	Ser	Cons	Exports	Imports		
		1	2	3	4	5	6	SUM(1:5) - (6)	
Ag	1	5	6	9	30	5	10	45	
Man	2	10	15	20	60	20	10	115	
Ser	3	12	13	5	110			140	
Wages	4	9	30	66				105	
Rents	5	9	51	40				100	
		<hr/>							
		45	115	140	200	25	-20		

Micro consistent: total value added (205 = 105+100) =
 consumption + trade surplus (205 = 200 + 20 - 10)

	<i>Net Ag</i>	<i>Net Man</i>	<i>Net Ser</i>	<i>Export</i>	<i>Imports</i>	<i>Welf</i>	<i>Cons</i>	<i>Row</i>
<i>Ag</i>	40	-6	-9	-5	10	-30		0
<i>Man</i>	-10	100	-20	-20	10	-60		0
<i>Ser</i>	-12	-13	135			-110		0
<i>Wages</i>	-9	-30	-66				105	0
<i>Rents</i>	-9	-51	-40				100	0
<i>For ex</i>				25	-20		-5	0
<i>Welfare</i>						200	-200	0
<i>Col sums</i>	0	0	0	0	0	0	0	

\$OFFTEXT

PARAMETERS

PWA world price of agriculture

PWM world price of manufactures;

PWA = 1;

PWM = 1;

\$ONTEXT

\$MODEL:M10_4

\$SECTORS :

AGR

MAN

SER

IMAGR EXAGR

EXMAN IMMAN

WELFARE

\$COMMODITIES :

PAGR_D

PAGR_F

PMAN_D

PMAN_F

PSER

PL

PK

PFX

PWEL

\$CONSUMERS :

CONS

\$PROD:AGR s:1

O:PAGR_D Q:45

I:PAGR_D Q:5

I:PMAN_D Q:10

I:PSER Q:12

I:PL Q:9

I:PK Q:9

\$PROD:MAN s:1

O:PMAN_D Q:115

I:PAGR_D Q:6

I:PMAN_D Q:15

I:PSER Q:13

I:PL Q:30

I:PK Q:51

\$PROD:SER s:1

O:PSER Q:140

I:PAGR_D Q:9

I:PMAN_D Q:20

I:PSER Q:5

I:PL Q:66

I:PK Q:40

\$PROD:IMAGR

O:PAGR_F Q: 10

I:PFX Q:(10*PWA)

\$PROD:EXMAN

O:PFX Q:(20*PWM)

I:PMAN_D Q: 20

\$PROD:EXAGR

*O:PFX Q:(5*PWA)*

I:PAGR_D Q: 5

\$PROD:IMMAN

O:PMAN_F Q: 10

*I:PFX Q:(10*PWM)*

\$PROD:WELFARE s:1 a:5 b:5

O:PWEL Q:200

I:PAGR_D Q:20 a:

I:PAGR_F Q:10 a:

I:PMAN_D Q:50 b:

I:PMAN_F Q:10 b:

I:PSER Q:110

\$DEMAND:CONS

D:PWEL

E:PL Q:105

E:PK Q:100

E:PFX Q:(-5)

\$OFFTEXT

```
$SYSINCLUDE MPSGESET M10_4
```

```
PWEL.FX = 1;
```

```
EXAGR.L = 1;
```

```
M10_4.ITERLIM = 0;
```

```
$INCLUDE M10_4.GEN
```

```
SOLVE M10_4 USING MCP;
```

```
M10_4.ITERLIM = 2000;
```

```
$INCLUDE M10_4.GEN
```

```
SOLVE M10_4 USING MCP;
```

```
* counterfactual: double the world price of the export good M
```

```
PWM = 2;
```

```
$INCLUDE M10_4.GEN
```

```
SOLVE M10_4 USING MCP;
```

```
* counterfactual: double the world price of the import good A
```

```
PWM = 1;
```

```
PWA = 2;
```

```
$INCLUDE M10_4.GEN
```

SOLVE M10_4 USING MCP;

The second techniques, shown in M10-4c, is to dampen responses by assuming that a share of one or more factors of production used in each sector is sector-specific.

A sector-specific factor is a factor that has no alternative use outside of its sector.

In model M10-4c, we create three addition commodity names: PKA, PKM, AND PKS for sector-specific capital in the three sectors.

The model assumes that specific capital is 60% of the capital initially observed to be used in each sector.

Note again how this dampens the responses to price change.

\$TITLE: M10-4c.gms

*CALIBRATES MODEL TO SHEET IO-1 IN M10-IOTABLE.XLS

*TRADE SURPLUS OF 5

*NETS OUT CROSS-HAULING: domestic and foreign goods perfect substitutes

*assumes a portion of capital in each sector is sector specific

*uses net trade, no Armington assumption

\$ONTEXT

		Ag	Man	Ser	Cons	Exports	Imports	
		1	2	3	4	5	6	UM(1:5) - (6)
Ag	1	5	6	9	30		5	45
Man	2	10	15	20	60	10		115
Ser	3	12	13	5	110			140
Wages	4	9	30	66				105
Rents	5	9	51	40				100
		45	115	140	200	10	-5	

Micro consistent: total value added (205 = 105+100) =
consumption + trade surplus (205 = 200 + 10 - 5)

	<i>Net Ag</i>	<i>Net Man</i>	<i>Net Ser</i>	<i>Export</i>	<i>Imports</i>	<i>Welf</i>	<i>Cons</i>	<i>Row</i>
<i>Ag</i>	40	-6	-9		5	-30		0
<i>Man</i>	-10	100	-20	-10		-60		0
<i>Ser</i>	-12	-13	135			-110		0
<i>Wages</i>	-9	-30	-66				105	0
<i>Rents</i>	-9	-51	-40				100	0
<i>For ex</i>				10	-5		-5	0
<i>Welfare</i>						200	-200	0
<i>Col sums</i>	0	0	0	0	0	0	0	

\$OFFTEXT

PARAMETERS

PWA world price of agriculture

PWM world price of manufactures

SHS share of sector-specific capital in total capital;

PWA = 1;

PWM = 1;

SHS = 0.6;

\$ONTEXT

\$MODEL:M10_4

\$SECTORS :

AGR

MAN

SER

IMAGR EXAGR

EXMAN IMMAN

WELFARE

\$COMMODITIES :

PAGR

PMAN

PSER

PL

PK

PKA PKM PKS

PFX

PWEL

\$CONSUMERS :

CONS

\$PROD:AGR s:1

O:PAGR Q:45

I:PAGR Q:5

I:PMAN Q:10

I:PSER Q:12

I:PL *Q:9*
I:PK *Q:(9*(1-SHS))*
I:PKA *Q:(9*SHS)*

\$PROD:MAN *s:1*

O:PMAN *Q:115*
I:PAGR *Q:6*
I:PMAN *Q:15*
I:PSER *Q:13*
I:PL *Q:30*
I:PK *Q:(51*(1-SHS))*
I:PKM *Q:(51*SHS)*

\$PROD:SER *s:1*

O:PSER *Q:140*
I:PAGR *Q:9*
I:PMAN *Q:20*
I:PSER *Q:5*
I:PL *Q:66*
I:PK *Q:(40*(1-SHS))*
I:PKS *Q:(40*SHS)*

\$PROD:IMAGR

O:PAGR *Q:5*
I:PFX *Q:(5*PWA)*

\$PROD: EXMAN

O: PFX Q: (10 * PWM)

I: PMAN Q: 10

\$PROD: EXAGR

O: PFX Q: (5 * PWA)

I: PAGR Q: (5 * 1.001)

\$PROD: IMMAN

O: PMAN Q: (10 * 0.999)

I: PFX Q: (10 * PWM)

\$PROD: WELFARE s: 1

O: PWEL Q: 200

I: PAGR Q: 30

I: PMAN Q: 60

I: PSER Q: 110

\$DEMAND: CONS

D: PWEL

E: PL Q: 105

E: PK Q: (100 * (1 - SHS))

E: PKA Q: (9 * SHS)

E: PKM Q: (51 * SHS)

E: PKS Q: (40 * SHS)

E: PFX Q: (-5)

\$OFFTEXT

\$SYSINCLUDE MPSEGET M10_4

PWEL.FX = 1;

M10_4.ITERLIM = 0;

\$INCLUDE M10_4.GEN

SOLVE M10_4 USING MCP;

M10_4.ITERLIM = 2000;

\$INCLUDE M10_4.GEN

SOLVE M10_4 USING MCP;

** counterfactual: double the world price of the export good M*

PWM = 2;

\$INCLUDE M10_4.GEN

SOLVE M10_4 USING MCP;

** counterfactual: double the world price of the import good A*

PWM = 1;

PWA = 2;

```
$INCLUDE M10_4.GEN  
SOLVE M10_4 USING MCP;
```

10.5 A More Complete IO Calibration Example using Sets

The (real) data for this exercise is shown in the Table below. It is micro-consistent, though there is a trade surplus in the benchmark. All *domestic* prices are assumed equal to one.

As before, all imports are assumed to go into final consumption, a totally unrealistic assumption.

Good IO data has a second table, which shown the sector distribution of imports into intermediate and final use.

The model features two features used earlier: first, a portion of all factors in each sector is sector specific (parameter SHARE).

Sample calibration problem from micro-consistent data

		Agriculture, fishery and energy	Mineral and metal products	Other industry	Business services	Government services	Household consumption	Government consumption	Gross investments	Exports	Imports	Import tariffs	sum 1-9 minus (10+ 11)
		1	2	3	4	5	6	7	8	9	10	11	
Agriculture, fishery and energy	1	19.7	24.9	76.0	19.2	13.0	71.2	0.0	8.4	10.5	14.7	6.0	222.2
Mineral and metal products	2	7.9	124.9	187.5	15.9	20.3	39.4	0.0	23.6	153.6	55.2	2.1	515.8
Other industry	3	19.6	29.5	311.8	129.8	63.7	296.5	0.0	504.0	495.5	239.7	6.4	1604.3
Business services	4	37.4	105.3	317.1	723.2	143.0	1002.3	21.4	87.6	141.0	75.0	30.2	2473.1
Government services	5	12.4	8.6	18.7	57.1	264.4	188.7	755.8	7.2	4.4	36.9	32.5	1247.9
Wage payments	6	60.3	167.2	508.8	680.1	556.7							
Gross operating surplus	7	75.2	50.7	175.7	821.4	202.1							
Production taxes less subsidies	8	-10.3	4.7	8.7	26.4	-15.3							
Value of production	9	222.2	515.8	1604.3	2473.1	1247.9							

column 1	row 1	household consumption	1598.1
19.7	19.7	plus gov consump	777.2
7.9	24.9	plus cap investment	630.8
19.6	76.0	plus trade surplus	383.5
37.4	19.2	minus wage income	-1973.1
12.4	13.0	minus capital income	-1325.1
60.3	71.2	minus prod taxes/sub	-14.2
75.2	0.0	minus import tariffs	-77.2
-10.3	8.4		
	10.5		0.0
	-14.7	Entered as <i>minus</i>	
	-6.0	their values in the IO table	
222.2	222.2		

Second, there is an Armington nesting structure between domestic and foreign final goods.

ARM(I) is the aggregator, with an elasticity of substitution of 2.

```
$PROD:ARM(I)  s:2  
O:PARM(I)  Q:(DCONS(I)+FCONS(I))  
I:PX(I)    Q:DCONS(I)  
I:PXF(I)   Q:FCONS(I)
```

Then welfare is produced in the upper nest, with an elasticity of substitution of one among the sectoral Armington aggregates.

```
$PROD:WEL  s:1  
O:PW      Q:(SUM(I, DCONS(I)+FCONS(I)))  
I:PARM(I) Q:(DCONS(I)+FCONS(I))
```

Counter-factual experiment is free trade.

\$TITLE:M10-5.GMS CALIBRATION EXERCISE, FROM SHEET IO-2, M10-IOTABLE.XLS

**CALIBRATES MODEL TO SHEET IO-2 IN M10-IOTABLE.XLS*

**assumes 10% of factors are sector specific to prevent "flats" problem*

**assumes domestic and foreign goods are Armington substitutes, sigma = 5*

**assume foreign goods only used for consumption, not intermediate usage*

**aggregates household, government, and investment demand to single consumer*

SETS R rows of the IO table /1*8/
C columns of the IO table /1*11/;

SETS RS(R) subset of rows for production sectors /1*5/
CS(C) subset of columns for production sectors /1*5/;

SETS RV(R) subset of rows for value added /6*7/
CD(C) subset of columns for final demand /6*8/;

SETS I allows switching of rows and columns in sectors /1*5/;

PARAMETERS

IO(RS,I) extracts intermediate use for \$prod blocks
VA(RV,I) extracts factor requirements for \$prod blocks
TAX(I) computes implied tax rates assuming output taxes
VALUE(I) value of sector I's output at consumer prices
PRODQ(I) output quantity = value (consumer prices = 1)

PRODP(I) producer prices calculated from consumer prices (=1)+ taxes
 PRODR(I) producer revenue: $prodp * prodq$
 COST(I) cost of all inputs to the sector: should equal prodr
 DCONS(I) final demand (household + government + investment demand)
 FCONS(I) foreign goods demand including tariffs (domestic prices = 1)
 EX(I) exports of sector i
 TAR(I) implied tariff rates on foreign goods
 PIM(I) implied foreign prices: $1 = pim * (1 + tar)$
 TBAL trade balance: exports minus imports
 SHARE share of each factor in each sector that is sector specific;

SHARE = 0.1;

TABLE BENCH(*,*)

	1	2	3	4	5
1	19.7	24.9	76.0	19.2	13.0
2	7.9	124.9	187.5	15.9	20.3
3	19.6	29.5	311.8	129.8	63.7
4	37.4	105.3	317.1	723.2	143.0
5	12.4	8.6	18.7	57.1	264.4
6	60.3	167.2	508.8	680.1	556.7
7	75.2	50.7	175.7	821.4	202.1
8	-10.3	4.7	8.7	26.4	-15.3
+					

	6	7	8	9	10	11
1	71.2	0.0	8.4	10.5	14.7	6.0
2	39.4	0.0	23.6	153.6	55.2	2.1
3	296.5	0.0	504.0	495.5	239.7	6.4
4	1002.3	21.4	87.6	141.0	75.0	30.2
5	188.7	755.8	7.2	4.4	36.9	32.5
6	0					
7	0					
8	0					

;

»

DISPLAY BENCH;

IO(RS,I) = BENCH(RS,I);

VA(RV,I) = BENCH(RV,I);

DISPLAY IO, VA;

VALUE(I) = **SUM**(RS, BENCH(RS, I)) + **SUM**(RV, BENCH(RV, I)) + BENCH("8", I);

TAX(I) = BENCH("8", I)/VALUE(I);

DISPLAY VALUE, TAX;

PRODQ(I) = VALUE(I);

PRODP(I) = 1 - TAX(I);

PRODR(I) = PRODQ(I)*PRODP(I);

$COST(I) = \text{SUM}(RS, \text{BENCH}(RS, I)) + \text{SUM}(RV, \text{BENCH}(RV, I));$

DISPLAY PRODQ, PRODP, PRODR, COST;

$DCONS(I) = \text{SUM}(CD, \text{BENCH}(I, CD)) - \text{BENCH}(I, "10") - \text{BENCH}(I, "11");$

$FCONS(I) = \text{BENCH}(I, "10") + \text{BENCH}(I, "11");$

$EX(I) = \text{BENCH}(I, "9");$

DISPLAY DCONS, FCONS, EX;

$TAR(I) = \text{BENCH}(I, "11") / FCONS(I);$

$PIM(I) = 1 / (1 + TAR(I));$

$TBAL = \text{SUM}(I, EX(I) - (FCONS(I) * PIM(I)));$

DISPLAY TAR, PIM;

\$ONTEXT

\$MODEL: IOCAL

\$SECTORS:

$X(I)$ *!domestic production of good i*

$E(I)$ *!exports of good i*

$M(I)$ *!imports of good i*

$ARM(I)$ *!Armington aggregator of domest (X) and foreign (M) good i*

WEL *!welfare*

\$COMMODITIES:

PX(I) !*price of domestic good i*
PXF(I) !*price of foreign good i*
PFX !*price of "foreign exchange"*
PF(RV) !*price of factor rv (mobile factors)*
PFS(RV,I) !*price of specific factor rv in sector i*
PARM(I) !*price of the Armington aggregate good i*
PW !*real consumer price index*

\$CONSUMERS:

CONS !*representative consumer*

\$PROD:X(I) s:1

O:PX(I) *Q:PRODQ(I)* *P:PRODP(I)* *A:CONS* *T:TAX(I)*
I:PX(RS) *Q:IO(RS,I)* *P:1*
I:PF(RV) *Q:(VA(RV,I)*(1-SHARE))* *P:1*
I:PFS(RV,I) *Q:(VA(RV,I)*SHARE)* *P:1*

\$PROD:E(I)

O:PFX *Q:EX(I)* *P:1*
I:PX(I) *Q:EX(I)* *P:1*

\$PROD:M(I)

O:PXF(I) *Q:FCONS(I)*
I:PFX *Q:(FCONS(I)*PIM(I))* *A:CONS* *T:TAR(I)*

\$PROD:ARM(I) s:2

O:PARM(I) Q:(DCONS(I)+FCONS(I))

I:PX(I) Q:DCONS(I)

I:PIX(I) Q:FCONS(I)

\$PROD:WEL s:1

O:PW Q:(SUM(I,DCONS(I)+FCONS(I)))

I:PARM(I) Q:(DCONS(I)+FCONS(I))

\$DEMAND:CONS

D:PW Q:(SUM(I,DCONS(I)+FCONS(I)))

E:PF(RV) Q:(SUM(I,VA(RV,I))*(1-SHARE))

E:PFS(RV,I) Q:(VA(RV,I)*(SHARE))

E:PFX Q:(-TBAL)

\$OFFTEXT

\$SYSINCLUDE MPSEGET IOCAL

PW.FX = 1;

IOCAL.ITERLIM = 0;

\$INCLUDE IOCAL.GEN

SOLVE IOCAL USING MCP;

**perturbation: check that calibrated solution is indeed an equilibrium*

X.L("2") = 2;

IOCAL.ITERLIM = 5000;
\$INCLUDE IOCAL.GEN
SOLVE IOCAL USING MCP;

**counterfactual: abolish all taxes*

TAX(I) = 0;
TAR(I) = 0;

\$INCLUDE IOCAL.GEN
SOLVE IOCAL USING MCP;

PARAMETER

OUT(I);

OUT(I) = X.L(I);

DISPLAY OUT;

\$LIBINCLUDE XLDUMP OUT M10-IOTABLE.XLS SHEET4!A3